# Maintainer: Alexey Pavlov <alexpux@gmail.com>

_realname=mako
pkgbase="python-${_realname}"
pkgname=("python-${_realname}")
pkgver=1.1.3
pkgrel=1
pkgdesc="A super-fast templating language that borrows the best ideas from the existing templating languages"
arch=('i686' 'x86_64')
url="https://pypi.python.org/pypi/Mako"
license=('MIT')
depends=("python-markupsafe"
         "python-beaker")
makedepends=("python-setuptools")
checkdepends=("python-nose")
provides=("python3-${_realname}")
conflicts=("python3-${_realname}")
replaces=("python3-${_realname}")
options=('staticlibs')
source=("https://pypi.python.org/packages/source/M/Mako/Mako-${pkgver}.tar.gz")
sha256sums=('8195c8c1400ceb53496064314c6736719c6f25e7479cd24c77be3d9361cddc27')

build() {
  cd "${srcdir}/Mako-${pkgver}"
  /usr/bin/python setup.py build
}

check() {
  cd "${srcdir}/Mako-${pkgver}"
  /usr/bin/python setup.py test
}

package() {
  cd "${srcdir}/Mako-${pkgver}"
  /usr/bin/python setup.py install --prefix=/usr --root="${pkgdir}" -O1
  install -Dm644 LICENSE "${pkgdir}/usr/share/licenses/python-${_realname}/COPYING"
}
